Solving for c_{in}, $j_{\text{in}}(=D\nabla c_{\text{in}})$ inside an active droplet with first order chemical reactions. Solving for c_{out}, $j_{\text{out}}(=D\nabla c_{\text{out}})$ outside the droplet for an active/passive droplet with first order chemical reactions.

____

Note: Cells highlighted in green indicate relevant parts implemented in the code.

____

Inside the droplet:

We solve the steady state reaction-diffusion equation: $D\nabla^2 c_{\text{in}} + \text{KF}\,(1 - c_{\text{in}}) - \text{KB}\,(c_{\text{in}}) = 0$ analytically for c_{in} for 1, 2 and 3 dimensions.

This reaction flux scheme is specific to the first order reactions which convert droplet to background field inside the droplet with rates KF and KB.

(Note that for the passive case, $c_{\text{in}} = c_{\text{in}}^{\text{eq}}$ and hence the local fluxes evaluated at the droplet surface $j_{\text{in}} = 0$).

We integrate $(-D\nabla c_{\text{in}})$ at r = R over the surface of the droplet to get the integrated flux J_{in}.

Finally, the mean flux inside the droplet is calculated as J_{in}/(Volume of the droplet), which is used as an input to 'reaction_inside' in the code.

____

———

Outside the droplet:

We solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{out} +$ KF $(1 - c_{out})$ - KB $(c_{out}) =$ 0 analytically for c_{out} for 1, 2 and 3 dimensions.

This reaction flux scheme is specific to the first order reactions which convert droplet to background field inside the droplet with rates KF and KB.

We integrate (- $D\nabla c_{out}$) at r = R over the surface of the droplet to get the integrated flux J_{out}.

3D droplets:

Solve the stationary state equation inside the droplets:

```
In[∘]:= ClearAll["Global`*"]

In[∘]:= $Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
           cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;

In[∘]:= sol3DInside[r_] = FullSimplify[FullSimplify[
          c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] +
                KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
             c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

$$\text{Out}[\circ]= \frac{\text{KF}\ r + (-\text{KF} + \text{cEqin}\ (\text{KB} + \text{KF}))\ R\ \text{Csch}\left[\frac{R}{\xi}\right]\ \text{Sinh}\left[\frac{r}{\xi}\right]}{(\text{KB} + \text{KF})\ r}$$

Check the boundary conditions

```
In[∘]:= FullSimplify@sol3DInside[R]
Out[∘]= cEqin

In[∘]:= Limit[D[sol3DInside[r], r], r → 0]
Out[∘]= 0
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[∘]:= FullSimplify[
         FullSimplify[Diffusion * Laplacian[sol3DInside[r], {r, ϕ, θ}, "Spherical"] +
            KF * (1 - sol3DInside[r]) - KB * (sol3DInside[r])] /. Diffusion → (KF + KB) * ξ^2]
Out[∘]= 0
```

Calculate surface integrated fluxes at the droplet surface

In[•]:=
```
TotalFluxInside3D =
  FullSimplify[FullSimplify[-4 π * R^2 * Diffusion * D[sol3DInside[r], r] /. r → R] /.
    Diffusion → (KF + KB) * ξ^2]
```

Out[•]= $4\,(-\mathrm{KF} + \mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\pi\,R\,\xi\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)$

In[•]:=
```
TotalFluxInside3DperVolume = FullSimplify[TotalFluxInside3D / ((4 / 3) π * R * R * R)]
```

Out[•]= $\frac{3\,(-\mathrm{KF} + \mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\xi\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)}{R^2}$

In[•]:=
```
LocalFluxInside3D = FullSimplify[-Diffusion * D[sol3DInside[r], r] /. r → R]
```

Out[•]= $\frac{\mathrm{Diffusion}\,(-\mathrm{KF} + \mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\left(\xi - R\,\mathrm{Coth}\left[\frac{R}{\xi}\right]\right)}{(\mathrm{KB}+\mathrm{KF})\,R\,\xi}$

Solve the stationary state equation outside the droplets:

In[•]:=
```
sol3DOutside[r_] = FullSimplify[Limit[FullSimplify[
     c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] +
           KF * (1 - c[r]) - KB * (c[r]), c[R] == cEqout, c[L] == cInf}, c, r] /.
      Diffusion → (KF + KB) * ξ^2], L → Infinity]]
```

••• Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[•]= $\frac{\mathrm{KF} + \frac{e^{\frac{-r+R}{\xi}}\,(-\mathrm{KF}+\mathrm{cEqout}\,(\mathrm{KB}+\mathrm{KF}))\,R}{r}}{\mathrm{KB}+\mathrm{KF}}$

Check the boundary conditions

In[•]:=
```
FullSimplify@sol3DOutside[R]
```

Out[•]= cEqout

In[•]:=
```
FullSimplify@sol3DOutside[Infinity]
```

Out[•]= $\frac{\mathrm{KF}}{\mathrm{KB}+\mathrm{KF}}$

Check the solution by plugging it in the Reaction-Diffusion equation

In[•]:=
```
FullSimplify[
  FullSimplify[Diffusion * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] +
     KF * (1 - sol3DOutside[r]) - KB * (sol3DOutside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[•]= 0

Calculate surface area integrated fluxes at the droplet surface

In[ ]:=
```
TotalFluxOutside3D =
 FullSimplify[FullSimplify[-4 π * R^2 * Diffusion * D[sol3DOutside[r], r] /. r → R]]
```

Out[ ]=
$$\frac{4\,\text{Diffusion}\,(-KF + cEqout\,(KB + KF))\,\pi\,R\,(R + \zeta)}{(KB + KF)\,\zeta}$$

Calculate stable droplet size for Active emulsions

In[ ]:=
```
LocalFluxOutside3D = FullSimplify[-Diffusion * D[sol3DOutside[r], r] /. r → R]
```

Out[ ]=
$$\frac{\text{Diffusion}\,(-KF + cEqout\,(KB + KF))\,(R + \zeta)}{(KB + KF)\,R\,\zeta}$$

In[ ]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
```

In[ ]:=
```
InterfacialSpeed[R_] := FullSimplify[
  FullSimplify[(LocalFluxInside3D - LocalFluxOutside3D) / (cEqin - cEqout)] /.
      cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ζ → Sqrt[1 / (KF + KB)]]
```

In[ ]:=
```
InterfacialSpeed[R]
```

Out[ ]=
$$\frac{0.998253 + 0.000953463\,R + (-0.00174732 - 0.00953463\,R)\,\text{Coth}\left[\frac{1}{100}\sqrt{\frac{11}{10}}\,R\right]}{R}$$

In[ ]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1}]
```

Out[ ]= {R → 1.83504}

In[ ]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 70}]
```

Out[ ]= {R → 68.6091}

In[ ]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[ ]= 1.8326

In[ ]:=
```
StableRadiusTheory = Sqrt[3 * Diffusion * (KF / (KB + KF)) / KB] /. Diffusion → 1
```

Out[ ]=
$$100\sqrt{\frac{3}{11}}$$

In[∘]:=
```
MeanReactionInside3D =
  Normal[Series[TotalFluxInside3DperVolume /. cEqin → (1 + 2 * γ / R) /.
     Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[∘]=
$$-\frac{1}{10000}-\frac{0.000018326}{R}+1.34391\times 10^{-10}\,R+\frac{11\,R^2}{15000000000}-1.4079\times 10^{-15}\,R^3-\frac{121\,R^4}{15750000000000000}$$

In[∘]:=
```
-KB
```

Out[∘]=
$$-\frac{1}{10000}$$

In[∘]:=

2D droplets:

Solve the stationary state equation inside the droplets:

In[∘]:=
```
ClearAll["Global`*"]
```

In[∘]:=
```
$Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
    cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;
```

In[∘]:=
```
sol2DInside[r_] =
  FullSimplify[FullSimplify[c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r],
          {r, ϕ}, "Polar"] + KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
        c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

Out[∘]=
$$\frac{\mathrm{KF}+\frac{(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselI}\left[0,\frac{r}{\xi}\right]}{\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]}}{\mathrm{KB}+\mathrm{KF}}$$

Check the boundary conditions

In[∘]:=
```
FullSimplify@sol2DInside[R]
```

Out[∘]=
```
cEqin
```

In[ ]:=
```
Limit[D[sol2DInside[r], r], r → 0]
```
Out[ ]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[ ]:=
```
FullSimplify[FullSimplify[Diffusion * Laplacian[sol2DInside[r], {r, ϕ}, "Polar"] +
    KF * (1 - sol2DInside[r]) - KB * (sol2DInside[r])] /. Diffusion → (KF + KB) * ξ^2]
```
Out[ ]= 0

Calculate surface integrated fluxes at the droplet surface

In[ ]:=
```
TotalFluxInside2D =
  FullSimplify[FullSimplify[-2 π * R * Diffusion * D[sol2DInside[r], r] /. r → R] /.
    Diffusion → (KF + KB) * ξ^2]
```
Out[ ]= $-\frac{2\,(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\pi\,\mathrm{R}\,\xi\,\mathrm{BesselI}\left[1,\frac{\mathrm{R}}{\xi}\right]}{\mathrm{BesselI}\left[0,\frac{\mathrm{R}}{\xi}\right]}$

In[ ]:=
```
TotalFluxInside2DperVolume = FullSimplify[TotalFluxInside2D / (π * R * R)]
```
Out[ ]= $-\frac{2\,(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\xi\,\mathrm{BesselI}\left[1,\frac{\mathrm{R}}{\xi}\right]}{\mathrm{R}\,\mathrm{BesselI}\left[0,\frac{\mathrm{R}}{\xi}\right]}$

In[ ]:=
```
LocalFluxInside2D = FullSimplify[-Diffusion * D[sol2DInside[r], r] /. r → R]
```
Out[ ]= $\frac{\mathrm{Diffusion}\,(\mathrm{KF}-\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselI}\left[1,\frac{\mathrm{R}}{\xi}\right]}{(\mathrm{KB}+\mathrm{KF})\,\xi\,\mathrm{BesselI}\left[0,\frac{\mathrm{R}}{\xi}\right]}$

Solve the stationary state equation outside the droplets:

In[ ]:=
```
sol2DOutsideComplex[r_] = FullSimplify[c[r] /. First@DSolve[
     {0 == Diffusion * Laplacian[c[r], {r, ϕ}, "Polar"] + KF * (1 - c[r]) - KB * (c[r]),
      c[R] == cEqout, c[L] == cInf}, c, r] /. Diffusion → (KF + KB) * ξ^2]
```
Out[ ]=
$$\left(\pi\left(\mathrm{BesselI}\left[0,\frac{\mathrm{R}}{\xi}\right]\left(\mathrm{KF}\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{L}}{\xi}\right]+(\mathrm{cInf}\,\mathrm{KB}+(-1+\mathrm{cInf})\,\mathrm{KF})\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{r}}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{\mathrm{L}}{\xi}\right]\left((\mathrm{KF}-\mathrm{cEqout}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{r}}{\xi}\right]-\mathrm{KF}\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{R}}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{\mathrm{r}}{\xi}\right]\left((\mathrm{cEqout}\,\mathrm{KB}+(-1+\mathrm{cEqout})\,\mathrm{KF})\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{L}}{\xi}\right]+(\mathrm{KF}-\mathrm{cInf}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{BesselY}\left[0,-\frac{i\,\mathrm{R}}{\xi}\right]\right)\right)\right)\Big/\left(2\,(\mathrm{KB}+\mathrm{KF})\left(-\mathrm{BesselI}\left[0,\frac{\mathrm{R}}{\xi}\right]\mathrm{BesselK}\left[0,\frac{\mathrm{L}}{\xi}\right]+\mathrm{BesselI}\left[0,\frac{\mathrm{L}}{\xi}\right]\mathrm{BesselK}\left[0,\frac{\mathrm{R}}{\xi}\right]\right)\right)$$

In[ ]:=
```
sol2DOutside[r_] =
  FullSimplify[sol2DOutsideComplex[r] /. BesselY[0, -I * x_] :→ -2 / π * BesselK[0, x]]
```

Out[ ]=
$$\left(\text{BesselI}\left[0, \frac{R}{\zeta}\right] \left(\text{KF BesselK}\left[0, \frac{L}{\zeta}\right] + (\text{cInf KB} + (-1 + \text{cInf})\ \text{KF})\ \text{BesselK}\left[0, \frac{r}{\zeta}\right]\right) + \text{BesselI}\left[0, \frac{L}{\zeta}\right] \left((\text{KF} - \text{cEqout}\ (\text{KB} + \text{KF}))\ \text{BesselK}\left[0, \frac{r}{\zeta}\right] - \text{KF BesselK}\left[0, \frac{R}{\zeta}\right]\right) + \text{BesselI}\left[0, \frac{r}{\zeta}\right] \left((\text{cEqout KB} + (-1 + \text{cEqout})\ \text{KF})\ \text{BesselK}\left[0, \frac{L}{\zeta}\right] + (\text{KF} - \text{cInf}\ (\text{KB} + \text{KF}))\ \text{BesselK}\left[0, \frac{R}{\zeta}\right]\right)\right) \Big/ \left((\text{KB} + \text{KF}) \left(\text{BesselI}\left[0, \frac{R}{\zeta}\right] \text{BesselK}\left[0, \frac{L}{\zeta}\right] - \text{BesselI}\left[0, \frac{L}{\zeta}\right] \text{BesselK}\left[0, \frac{R}{\zeta}\right]\right)\right)$$

Check the boundary conditions

In[ ]:=
```
FullSimplify@sol2DOutside[R]
```

Out[ ]= cEqout

In[ ]:=
```
FullSimplify@sol2DOutside[L]
```

Out[ ]= cInf

Check the solution by plugging it in the Reaction-Diffusion equation

In[ ]:=
```
FullSimplify[
  FullSimplify[Diffusion * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] +
     KF * (1 - sol2DOutside[r]) - KB * (sol2DOutside[r])] /. Diffusion → (KF + KB) * ζ^2]
```

Out[ ]= 0

Calculate surface area integrated fluxes at the droplet surface

In[ ]:=
```
TotalFluxOutside2D =
  FullSimplify[FullSimplify[-2 π * R * Diffusion * D[sol2DOutside[r], r] /. r → R]]
```

Out[ ]=
$$\left(2\ \text{Diffusion}\ \pi \left((-\text{KF} + \text{cInf}\ (\text{KB} + \text{KF}))\ \zeta + (\text{KF} - \text{cEqout}\ (\text{KB} + \text{KF}))\ R \left(\text{BesselI}\left[1, \frac{R}{\zeta}\right] \text{BesselK}\left[0, \frac{L}{\zeta}\right] + \text{BesselI}\left[0, \frac{L}{\zeta}\right] \text{BesselK}\left[1, \frac{R}{\zeta}\right]\right)\right)\right) \Big/ \left((\text{KB} + \text{KF})\ \zeta \left(\text{BesselI}\left[0, \frac{R}{\zeta}\right] \text{BesselK}\left[0, \frac{L}{\zeta}\right] - \text{BesselI}\left[0, \frac{L}{\zeta}\right] \text{BesselK}\left[0, \frac{R}{\zeta}\right]\right)\right)$$

Calculate stable droplet size for Active emulsions

In[∘]:=
```
LocalFluxOutside2D = FullSimplify[-Diffusion * D[sol2DOutside[r], r] /. r → R]
```

Out[∘]=

$$\left(\text{Diffusion}\left(\frac{(-\text{KF}+\text{cInf}\,(\text{KB}+\text{KF}))\,\zeta}{R}+(\text{KF}-\text{cEqout}\,(\text{KB}+\text{KF}))\left(\text{BesselI}\left[1,\frac{R}{\zeta}\right]\text{BesselK}\left[0,\frac{L}{\zeta}\right]+\text{BesselI}\left[0,\frac{L}{\zeta}\right]\text{BesselK}\left[1,\frac{R}{\zeta}\right]\right)\right)\right)\Big/\left((\text{KB}+\text{KF})\,\zeta\left(\text{BesselI}\left[0,\frac{R}{\zeta}\right]\text{BesselK}\left[0,\frac{L}{\zeta}\right]-\text{BesselI}\left[0,\frac{L}{\zeta}\right]\text{BesselK}\left[0,\frac{R}{\zeta}\right]\right)\right)$$

In[∘]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
L = 100 * R;
```

In[∘]:=
```
InterfacialSpeed[R_] :=
 FullSimplify[(LocalFluxInside2D - LocalFluxOutside2D) / (cEqin - cEqout)] /.
     cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
   Diffusion → 1 /. ζ → Sqrt[1 / (KF + KB)]
```

In[∘]:=
```
InterfacialSpeed[R]
```

Out[∘]=

$$\left(0.000953463\times\left(-11.\,\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\text{BesselI}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\text{BesselK}\left[0,\sqrt{\frac{11}{10}}R\right]+\text{BesselI}\left[0,\sqrt{\frac{11}{10}}R\right]\left(\left(-1+11\times\left(1+\frac{0.1666}{R}\right)\right)\text{BesselI}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\text{BesselK}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]+\left(-1+\frac{1.8326}{R}\right)\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\text{BesselK}\left[1,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\right)\right)\right)\Big/\left(\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\left(-\text{BesselI}\left[0,\sqrt{\frac{11}{10}}R\right]\text{BesselK}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]+\text{BesselI}\left[0,\frac{1}{100}\sqrt{\frac{11}{10}}R\right]\text{BesselK}\left[0,\sqrt{\frac{11}{10}}R\right]\right)\right)$$

In[∘]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1}]
```

Out[∘]=
```
{R → 1.84788}
```

In[∘]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 40}]
```

Out[∘]=
```
{R → 36.6037}
```

In[∘]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[∘]=
```
1.8326
```

In[ ]:=
```
MeanReactionInside2D =
 Normal[Series[TotalFluxInside2DperVolume /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[ ]=
$$-\frac{1}{10000}-\frac{0.000018326}{R}+2.51983\times10^{-10}\,R+\frac{11\,R^2}{8000000000}-4.61968\times10^{-15}\,R^3-\frac{121\,R^4}{4800000000000000}$$

In[ ]:=
```
- KB
```

Out[ ]=
$$-\frac{1}{10000}$$

In[ ]:=

1D droplets:

Solve the stationary state equation inside the droplets:

In[ ]:=
```
ClearAll["Global`*"]
```

In[ ]:=
```
$Assumptions = Diffusion > 0 && KF > 0 && KB > 0 &&
   cEqin > 0 && cEqout > 0 && cInf > 0 && R > 0 && ξ > 0 && γ > 0 && L > R;
```

In[ ]:=
```
sol1DInside[r_] = FullSimplify[FullSimplify[
    c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r}, "Cartesian"] +
          KF * (1 - c[r]) - KB * (c[r]), Derivative[1][c][0] == 0,
        c[R] == cEqin}, c, r]] /. Diffusion → (KF + KB) * ξ^2]
```

Out[ ]=
$$\frac{KF+\left(-KF+cEqin\,(KB+KF)\right)\,Cosh\left[\frac{r}{\xi}\right]\,Sech\left[\frac{R}{\xi}\right]}{KB+KF}$$

Check the boundary conditions

In[ ]:=
```
FullSimplify@sol1DInside[R]
```

Out[ ]=
```
cEqin
```

In[ ]:=
```
Limit[D[sol1DInside[r], r], r → 0]
```
Out[ ]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[ ]:=
```
FullSimplify[
 FullSimplify[Diffusion * Laplacian[sol1DInside[r], {r}, "Cartesian"] +
    KF * (1 - sol1DInside[r]) - KB * (sol1DInside[r])] /. Diffusion → (KF + KB) * ξ^2]
```
Out[ ]= 0

Calculate surface integrated fluxes at the droplet surface

In[ ]:=
```
TotalFluxInside1D =
 FullSimplify[FullSimplify[-2 * Diffusion * D[sol1DInside[r], r] /. r → R] /.
   Diffusion → (KF + KB) * ξ^2]
```
Out[ ]= $-2\,(-\mathrm{KF}+\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\xi\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]$

In[ ]:=
```
TotalFluxInside1DperVolume = FullSimplify[TotalFluxInside1D / (2 * R)]
```
Out[ ]= $\frac{(\mathrm{KF}-\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\xi\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]}{R}$

In[ ]:=
```
LocalFluxInside1D = FullSimplify[-Diffusion * D[sol1DInside[r], r] /. r → R]
```
Out[ ]= $\frac{\mathrm{Diffusion}\,(\mathrm{KF}-\mathrm{cEqin}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{Tanh}\left[\frac{R}{\xi}\right]}{(\mathrm{KB}+\mathrm{KF})\,\xi}$

Solve the stationary state equation outside the droplets:

In[ ]:=
```
sol1DOutside[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diffusion * Laplacian[c[r], {r}, "Cartesian"] +
        KF * (1 - c[r]) - KB * (c[r]), c[R] == cEqout,
      c[L] == cInf}, c, r] /. Diffusion → (KF + KB) * ξ^2]
```
Out[ ]= $-\frac{1}{\left(-e^{\frac{2L}{\xi}}+e^{\frac{2R}{\xi}}\right)(\mathrm{KB}+\mathrm{KF})}\,2\,e^{\frac{L+R}{\xi}}\left((-\mathrm{KF}+\mathrm{cEqout}\,(\mathrm{KB}+\mathrm{KF}))\,\mathrm{Sinh}\left[\frac{L-r}{\xi}\right]+\mathrm{KF}\,\mathrm{Sinh}\left[\frac{L-R}{\xi}\right]+(\mathrm{cInf}\,\mathrm{KB}+(-1+\mathrm{cInf})\,\mathrm{KF})\,\mathrm{Sinh}\left[\frac{r-R}{\xi}\right]\right)$

Check the boundary conditions

In[ ]:=
```
FullSimplify@sol1DOutside[R]
```
Out[ ]= cEqout

In[ ]:=
```
FullSimplify@sol1DOutside[L]
```
Out[ ]= cInf

Check the solution by plugging it in the Reaction-Diffusion equation

In[ ]:=
```
FullSimplify[
 FullSimplify[Diffusion * Laplacian[sol1DOutside[r], {r}, "Cartesian"] +
    KF * (1 - sol1DOutside[r]) - KB * (sol1DOutside[r])] /. Diffusion → (KF + KB) * ξ^2]
```

Out[ ]= 0

Calculate surface area integrated fluxes at the droplet surface

In[ ]:=
```
TotalFluxOutside1D = FullSimplify[-2 * Diffusion * D[sol1DOutside[r], r] /. r → R]
```

Out[ ]=
$$\frac{4\,\text{Diffusion}\,e^{\frac{L+R}{\xi}}\left(-KF+cInf\,(KB+KF)+(KF-cEqout\,(KB+KF))\,\text{Cosh}\left[\frac{L-R}{\xi}\right]\right)}{\left(-e^{\frac{2L}{\xi}}+e^{\frac{2R}{\xi}}\right)(KB+KF)\,\xi}$$

Calculate stable droplet size for Active emulsions

In[ ]:=
```
LocalFluxOutside1D =
 FullSimplify[FullSimplify[-Diffusion * D[sol1DOutside[r], r] /. r → R]]
```

Out[ ]=
$$\frac{2\,\text{Diffusion}\,e^{\frac{L+R}{\xi}}\left(-KF+cInf\,(KB+KF)+(KF-cEqout\,(KB+KF))\,\text{Cosh}\left[\frac{L-R}{\xi}\right]\right)}{\left(-e^{\frac{2L}{\xi}}+e^{\frac{2R}{\xi}}\right)(KB+KF)\,\xi}$$

In[ ]:=
```
KF = (1*^-5);
KB = (1*^-4);
γ = 0.0833;
cInf = KF / (KF + KB);
L = 100 * R;
```

In[ ]:=
```
InterfacialSpeed[R_] := FullSimplify[
  FullSimplify[(LocalFluxInside1D - LocalFluxOutside1D) / (cEqin - cEqout)] /.
      cEqout → (2 * γ / R) /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)]]
```

In[ ]:=
```
CriticalRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 1.83}]
```

Out[ ]= {R → 2.56409}

In[ ]:=
```
StableRadiusNumerical = FindRoot[InterfacialSpeed[R], {R, 7}]
```

Out[ ]= {R → 6.78963}

In[ ]:=
```
CriticalRadiusTheory = 2 * γ / cInf
```

Out[ ]= 1.8326

In[∘]:=
```
MeanReactionInside1D =
 Normal[Series[TotalFluxInside1DperVolume /. cEqin → (1 + 2 * γ / R) /.
    Diffusion → 1 /. ξ → Sqrt[1 / (KF + KB)], {R, 0, 4}]]
```

Out[∘]=
$$-\frac{1}{10000}-\frac{0.000018326}{R}+6.71953\times10^{-10}\,R+\frac{11\,R^{2}}{3000000000}-2.95659\times10^{-14}\,R^{3}-\frac{121\,R^{4}}{750000000000000}$$

In[∘]:=
```
-KB
```

Out[∘]=
$$-\frac{1}{10000}$$