Solving for c_{in}, $j_{\text{in}}(= D\nabla c_{\text{in}})$ inside an active droplet with linearized chemical reactions for any generic reaction flux s(c_{in}). Solving for c_{out}, $j_{\text{out}}(= D\nabla c_{\text{out}})$ inside the shell for an active/passive droplet with linearized chemical reactions for any generic reaction flux s(c_{out}).

____

Note: Cells highlighted in green indicate relevant parts implemented in the code.

____

Inside the droplet:

We aim to solve the steady state reaction-diffusion equation: $D\nabla^2 c_{\text{in}} + \text{s}(c_{\text{in}}) = 0$ for any generic reaction flux scheme s(c_{in}) inside the droplet. (Note that for the passive case, $c_{\text{in}} = c_{\text{in}}^{\text{eq}}$ and hence $j_{\text{in}} = 0$).

We can linearize the generic reaction flux s(c_{in}) around c_{in}^0 as:

$\text{s}(c_{\text{in}}) \approx \text{s}(c_{\text{in}}^0) - \text{s}`(c_{\text{in}}^0)(c_{\text{in}} - c_{\text{in}}^0) \approx \text{s}(c_{\text{in}}^0) - k_{\text{in}}(c_{\text{in}} - c_{\text{in}}^0)$

Denoting s(c_{in}^0) as ScZeroIn and k_{in} as k, where k>0 (Refer to Review, *Eqs*. 4.15).

We then solve $D\nabla^2 c_{\text{in}} +$ ScZeroIn - k (c_{in} - cZeroIn) = 0 analytically for c_{in} for 1, 2 and 3 dimensions.

We integrate (- D∇c_{in}) at r = R over the surface of the droplet to get the integrated flux J_{in}.

Finally, the mean flux inside the droplet is calculated as J_{in}/(Volume of the droplet), which is used as an input to ‘reaction_inside’ in the code.

——

Outside the droplet = inside the shell:

We aim to solve the steady state reaction-diffusion equation: $D\,\nabla^2 c_{out}$ + s(c_{out}) = 0 for any generic reaction flux scheme s(c_{out}) inside each shell sector. (Note that for the passive case, we only solve $D\,\nabla^2 c_{out} = 0$).

Generally, as the shell thickness L is a simulation parameter (and can be large or small as compared to the droplet radius R), c_{out} inside the shell sector can vary a lot spatially. Hence, as a first approximation, we assume the reaction flux s(c_{out}) to be a linear function of c_{out}.

We then approximate s(c_{out}) as s(c_{out}) ≈ A - k c_{out}, where k > 0 (Refer to Review, Eqs. 4.15)

Note that A - k c_{out} is used as an input to ‘reaction_outside’ in the code.

We solve for A and k from the following two equations:

1. s(c_{out}) at r = R: s(c_{out}^{eq}) = A - k c_{out}^{eq}

2. s(c_{out}) at r = R + L: s(cfar) = A - k c_{far}

After determining A and k, we solve $D\,\nabla^2 c_{out}$ + A - k c_{out} = 0 analytically for c_{out} 1, 2 and 3 dimensions. We then integrate (- D∇c_{out}) at r = R over the surface of the droplet to get the integrated flux J_{out}.

3D: Inside the droplet:

```
In[1]:= ClearAll["Global`*"]
```

```
In[2]:= $Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 3D droplets

In[3]:=
```
sol3DInside[r_] =
  FullSimplify[FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[
          c[r], {r, ϕ, θ}, "Spherical"] + ScZeroIn - k * (c[r] - cZeroIn),
        Derivative[1][c][0] == 0, c[R] == cEqIn}, c, r]] /. Diff → k * ξ^2]
```

Out[3]= $$\text{cZeroIn} + \frac{\text{ScZeroIn}}{\text{k}} - \frac{\text{R}\,(-\text{cEqIn k} + \text{cZeroIn k} + \text{ScZeroIn})\,\text{Csch}\left[\frac{\text{R}}{\xi}\right]\,\text{Sinh}\left[\frac{\text{r}}{\xi}\right]}{\text{k r}}$$

Check the boundary conditions

In[4]:=
```
FullSimplify@sol3DInside[R]
```

Out[4]= cEqIn

In[5]:=
```
Limit[D[sol3DInside[r], r], r → 0]
```

Out[5]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[6]:=
```
FullSimplify[
  FullSimplify[Diff * Laplacian[sol3DInside[r], {r, ϕ, θ}, "Spherical"] +
     ScZeroIn - k * (sol3DInside[r] - cZeroIn)] /. Diff → k * ξ^2]
```

Out[6]= 0

Calculate surface area integrated fluxes at the droplet surface

In[7]:=
```
FluxInside3D = FullSimplify[
   FullSimplify[-4 π * R^2 * Diff * D[sol3DInside[r], r] /. r → R] /. Diff → k * ξ^2]
```

Out[7]= $$4\,\pi\,\text{R}\,(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})\,\xi\left(\xi - \text{R Coth}\left[\frac{\text{R}}{\xi}\right]\right)$$

In[8]:=
```
FluxInside3DperVolume [R_] = FullSimplify[FluxInside3D / ((4 / 3) π * R * R * R)]
```

Out[8]= $$\frac{3\,(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})\,\xi\left(\xi - \text{R Coth}\left[\frac{\text{R}}{\xi}\right]\right)}{\text{R}^2}$$

In[9]:=
```
Normal[Series[FluxInside3DperVolume[R], {R, 0, 4}]]
```

Out[9]= $$-\text{cEqIn k} + \text{cZeroIn k} + \text{ScZeroIn} - \frac{2\,\text{R}^4\,(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})}{315\,\xi^4} + \frac{\text{R}^2\,(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})}{15\,\xi^2}$$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[10]:=
```
temp[F_] =
  FullSimplify[FullSimplify[FluxInside3D / ((4 / 3) π * R * R * R)] /. R → F * ξ]
```

Out[10]= $-\frac{3\,(\text{cEqIn k} - \text{cZeroIn k} - \text{ScZeroIn})\,(-1 + \text{F Coth}[\text{F}])}{\text{F}^2}$

In[11]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```

Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[11]= ScZeroIn

3D: Outside the droplet = Inside the shell:

In[12]:=
```
ClearAll["Global`*"]
```

In[13]:=
```
$Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[13]= k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

In[14]:=
```
eq1 = A - k * cEqOut == sOutceqout;
eq2 = A - k * cfar == sOutcfar;
NSolve[{eq1, eq2}, {A, k}]
```

Out[16]= $\left\{\left\{\text{A} \to \frac{1.\times(1.\,\text{cfar sOutceqout} - 1.\,\text{cEqOut sOutcfar})}{-1.\,\text{cEqOut} + 1.\,\text{cfar}},\ \text{k} \to -\frac{1.\times(1.\,\text{sOutceqout} - 1.\,\text{sOutcfar})}{1.\,\text{cEqOut} - 1.\,\text{cfar}}\right\}\right\}$

In[17]:= $\text{A} \to \frac{(\text{cfar sOutceqout} - \text{cEqOut sOutcfar})}{-\text{cEqOut} + \text{cfar}}$

Out[17]= $\text{A} \to \frac{\text{cfar sOutceqout} - \text{cEqOut sOutcfar}}{-\text{cEqOut} + \text{cfar}}$

In[18]:= $\text{k} \to -\frac{(\text{sOutceqout} - \text{sOutcfar})}{\text{cEqOut} - \text{cfar}}$

Out[18]= $\text{k} \to -\frac{\text{sOutceqout} - \text{sOutcfar}}{\text{cEqOut} - \text{cfar}}$

Solve the stationary state equation for 3D droplets

In[19]:=
```
sol3DOutside[r_] = FullSimplify[c[r] /.
    First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] + A - k * c[r],
      c[R] == cEqOut, c[R + L] == cfar}, c, r] /. Diff → k * ξ^2]
```

Out[19]=
$$\frac{1}{k\,r} e^{L/\xi} \left(-1 + \mathrm{Coth}\left[\frac{L}{\xi}\right]\right) \left(A\, r\, \mathrm{Sinh}\left[\frac{L}{\xi}\right] - (A - \mathrm{cfar}\, k)\,(L + R)\,\mathrm{Sinh}\left[\frac{r - R}{\xi}\right] + (-A + \mathrm{cEqOut}\, k)\, R\, \mathrm{Sinh}\left[\frac{L - r + R}{\xi}\right]\right)$$

Check the boundary conditions

In[20]:=
```
FullSimplify@sol3DOutside[R]
```

Out[20]= cEqOut

In[21]:=
```
FullSimplify@sol3DOutside[R + L]
```

Out[21]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[22]:=
```
FullSimplify[Diff * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] +
   A - k * sol3DOutside[r]] /. Diff → k * ξ^2
```

Out[22]= 0

Calculate surface area integrated fluxes at the droplet surface

In[23]:=
```
FluxOutside3D =
  FullSimplify[FullSimplify[-4 π * R^2 * Diff * D[sol3DOutside[r], r] /. r → R]]
```

Out[23]=
$$\frac{4\, \mathrm{Diff}\, \pi\, R \left(-\left((A - \mathrm{cEqOut}\, k)\left(\xi + R\, \mathrm{Coth}\left[\frac{L}{\xi}\right]\right)\right) + (A - \mathrm{cfar}\, k)\,(L + R)\,\mathrm{Csch}\left[\frac{L}{\xi}\right]\right)}{k\, \xi}$$

3D: Outside the droplet = Inside the shell when ceqout ~ cfar (when B → 0):

In[24]:=
```
ClearAll["Global`*"]
```

In[25]:=
```
$Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[25]= cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 3D droplets

In[26]:=
```
sol3DOutside[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"] + A,
      c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[26]=
$$\frac{6\,\text{cfar}\,\text{Diff}\,(r-R)\,(L+R)+(L-r+R)\,(6\,\text{cEqOut}\,\text{Diff}\,R+A\,L\,(r-R)\,(L+r+2\,R))}{6\,\text{Diff}\,L\,r}$$

Check the boundary conditions

In[27]:=
```
FullSimplify@sol3DOutside[R]
```

Out[27]= cEqOut

In[28]:=
```
FullSimplify@sol3DOutside[R + L]
```

Out[28]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[29]:=
```
FullSimplify[Diff * Laplacian[sol3DOutside[r], {r, ϕ, θ}, "Spherical"] + A]
```

Out[29]= 0

Calculate surface area integrated fluxes at the droplet surface

In[30]:=
```
FluxOutside3D =
 FullSimplify[FullSimplify[-4 π * R^2 * Diff * D[sol3DOutside[r], r] /. r → R]]
```

Out[30]=
$$-\frac{2\,\pi\,R\,\left(-6\,\text{cEqOut}\,\text{Diff}\,(L+R)+6\,\text{cfar}\,\text{Diff}\,(L+R)+A\,L^{2}\,(L+3\,R)\right)}{3\,L}$$

3D: Outside the droplet = Inside the shell (Passive droplet):

In[31]:=
```
ClearAll["Global`*"]
```

In[32]:=
```
$Assumptions = cEqOut > 0 && L > 0 && Diff > 0 && R > 0
```

Out[32]= $\text{cEqOut} > 0 \,\&\&\, L > 0 \,\&\&\, \text{Diff} > 0 \,\&\&\, R > 0$

Solve the stationary state equation for 3D droplets

In[33]:=
```
sol3DOutsidePassive[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ, θ}, "Spherical"],
      c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[33]=
$$\frac{\text{cfar}\,(r-R)\,(L+R)+\text{cEqOut}\,R\,(L-r+R)}{L\,r}$$

Check the boundary conditions

In[34]:=
```
FullSimplify@sol3DOutsidePassive[R]
```
Out[34]=
```
cEqOut
```

In[35]:=
```
FullSimplify@sol3DOutsidePassive[R + L]
```
Out[35]=
```
cfar
```

Check the solution by plugging it in the Diffusion equation

In[36]:=
```
FullSimplify[Diff * Laplacian[sol3DOutsidePassive[r], {r, ϕ, θ}, "Spherical"]]
```
Out[36]=
```
0
```

Calculate surface area integrated fluxes at the droplet surface

In[37]:=
```
FluxOutside3DPassive = FullSimplify[
  FullSimplify[-4 π * R^2 * Diff * D[sol3DOutsidePassive[r], r] /. r → R]]
```
Out[37]=

$$\frac{4\ (\text{cEqOut} - \text{cfar})\ \text{Diff}\ \pi\ R\ (L + R)}{L}$$

In[38]:=

In[39]:=

2D: Inside the droplet:

In[40]:=
```
ClearAll["Global`*"]
```

In[41]:=
```
$Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 2D droplets

In[42]:=
```
sol2DInside[r_] = FullSimplify[FullSimplify[c[r] /. First@DSolve[
    {0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + ScZeroIn - k * (c[r] - cZeroIn),
     Derivative[1][c][0] == 0, c[R] == cEqIn}, c, r]] /. Diff → k * (ξ^2)]
```

Out[42]=
$$\text{cZeroIn} + \frac{\text{ScZeroIn}}{k} + \frac{(\text{cEqIn}\,k - \text{cZeroIn}\,k - \text{ScZeroIn})\,\text{BesselI}\left[0, \frac{r}{\xi}\right]}{k\,\text{BesselI}\left[0, \frac{R}{\xi}\right]}$$

Check the boundary conditions

In[43]:=
```
FullSimplify@sol2DInside[R]
```

Out[43]= cEqIn

In[44]:=
```
Limit[D[sol2DInside[r], r], r → 0]
```

Out[44]= 0

Check the solution by plugging it in the Reaction-Diffusion equation

In[45]:=
```
FullSimplify[
 FullSimplify[Diff * Laplacian[sol2DInside[r], {r, ϕ}, "Polar"] + ScZeroIn -
   k * (sol2DInside[r] - cZeroIn)] /. Diff → k * (ξ^2)]
```

Out[45]= 0

Calculate surface area integrated fluxes at the droplet surface

In[46]:=
```
FluxInside2D = FullSimplify[
  FullSimplify[-2 π * R * Diff * D[sol2DInside[r], r] /. r → R] /. Diff → k * (ξ^2)]
```

Out[46]=
$$\frac{2\,\pi\,R\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{BesselI}\left[1, \frac{R}{\xi}\right]}{\text{BesselI}\left[0, \frac{R}{\xi}\right]}$$

In[47]:=
```
FluxInside2DperVolume[R_] = FullSimplify[FluxInside2D / (π * R * R)]
```

Out[47]=
$$\frac{2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{BesselI}\left[1, \frac{R}{\xi}\right]}{R\,\text{BesselI}\left[0, \frac{R}{\xi}\right]}$$

In[48]:=
```
Normal[Series[FluxInside2DperVolume[R], {R, 0, 4}]]
```

Out[48]=
$$\left(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn} + \frac{R^4\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{48\,\xi^4} - \frac{R^2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{8\,\xi^2}\right)$$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[49]:=
```
temp[F_] = FullSimplify[FullSimplify[FluxInside2D / (π * R * R)] /. R → F * ξ]
```

Out[49]=
$$\frac{2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{BesselI}[1, F]}{F\,\text{BesselI}[0, F]}$$

In[50]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```

Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[50]= ScZeroIn

2D: Outside the droplet = Inside the shell:

In[51]:=
```
ClearAll["Global`*"]
```

In[52]:=
```
$Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[52]= k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

In[53]:=
```
eq1 = A - k * cEqOut == sOutceqout;
eq2 = A - k * cfar == sOutcfar;
NSolve[{eq1, eq2}, {A, k}]
```

Out[55]=
$$\left\{\left\{A \to \frac{1.\times(1.\,\text{cfar sOutceqout} - 1.\,\text{cEqOut sOutcfar})}{-1.\,\text{cEqOut} + 1.\,\text{cfar}},\ k \to -\frac{1.\times(1.\,\text{sOutceqout} - 1.\,\text{sOutcfar})}{1.\,\text{cEqOut} - 1.\,\text{cfar}}\right\}\right\}$$

In[56]:=
$$A \to \frac{(\text{cfar sOutceqout} - \text{cEqOut sOutcfar})}{-\text{cEqOut} + \text{cfar}}$$

Out[56]=
$$A \to \frac{\text{cfar sOutceqout} - \text{cEqOut sOutcfar}}{-\text{cEqOut} + \text{cfar}}$$

In[57]:=
$$k \to -\frac{(\text{sOutceqout} - \text{sOutcfar})}{\text{cEqOut} - \text{cfar}}$$

Out[57]=
$$k \to -\frac{\text{sOutceqout} - \text{sOutcfar}}{\text{cEqOut} - \text{cfar}}$$

Solve the stationary state equation for 2D droplets

In[58]:=
```
solI2DOutside[r_] = FullSimplify[
  c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + A - k * c[r],
      c[R] == cEqOut, c[R + L] == cfar}, c, r] /. Diff → k * ξ^2]
```

Out[58]=
$$\left(\pi\left(\mathrm{BesselI}\left[0,\frac{L+R}{\xi}\right]\left((-A+\mathrm{cEqOut}\,k)\,\mathrm{BesselY}\left[0,-\frac{i\,r}{\xi}\right]+A\,\mathrm{BesselY}\left[0,-\frac{i\,R}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]\left((A-\mathrm{cfar}\,k)\,\mathrm{BesselY}\left[0,-\frac{i\,r}{\xi}\right]-A\,\mathrm{BesselY}\left[0,-\frac{i\,(L+R)}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{r}{\xi}\right]\left((-A+\mathrm{cfar}\,k)\,\mathrm{BesselY}\left[0,-\frac{i\,R}{\xi}\right]+(A-\mathrm{cEqOut}\,k)\,\mathrm{BesselY}\left[0,-\frac{i\,(L+R)}{\xi}\right]\right)\right)\right)\Big/\left(2\left(-k\,\mathrm{BesselI}\left[0,\frac{L+R}{\xi}\right]\mathrm{BesselK}\left[0,\frac{R}{\xi}\right]+k\,\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]\mathrm{BesselK}\left[0,\frac{L+R}{\xi}\right]\right)\right)$$

In[59]:=
```
sol2DOutside[r_] =
 FullSimplify[solI2DOutside[r] /. BesselY[0, -I * x_] :→ -2 / π * BesselK[0, x]]
```

Out[59]=
$$\left(\mathrm{BesselI}\left[0,\frac{L+R}{\xi}\right]\left((-A+\mathrm{cEqOut}\,k)\,\mathrm{BesselK}\left[0,\frac{r}{\xi}\right]+A\,\mathrm{BesselK}\left[0,\frac{R}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]\left((A-\mathrm{cfar}\,k)\,\mathrm{BesselK}\left[0,\frac{r}{\xi}\right]-A\,\mathrm{BesselK}\left[0,\frac{L+R}{\xi}\right]\right)+\mathrm{BesselI}\left[0,\frac{r}{\xi}\right]\left((-A+\mathrm{cfar}\,k)\,\mathrm{BesselK}\left[0,\frac{R}{\xi}\right]+(A-\mathrm{cEqOut}\,k)\,\mathrm{BesselK}\left[0,\frac{L+R}{\xi}\right]\right)\right)\Big/\left(k\,\mathrm{BesselI}\left[0,\frac{L+R}{\xi}\right]\mathrm{BesselK}\left[0,\frac{R}{\xi}\right]-k\,\mathrm{BesselI}\left[0,\frac{R}{\xi}\right]\mathrm{BesselK}\left[0,\frac{L+R}{\xi}\right]\right)$$

Check the boundary conditions

In[60]:=
```
FullSimplify[sol2DOutside[R]]
```

Out[60]= cEqOut

In[61]:=
```
FullSimplify[sol2DOutside[R + L]]
```

Out[61]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[62]:=
```
FullSimplify[Diff * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] +
   A - k * sol2DOutside[r]] /. Diff → k * ξ^2
```

Out[62]= 0

Calculate surface area integrated fluxes at the droplet surface

In[63]:=
```
FluxOutside2D =
 FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutside[r], r] /. r → R]]
```

Out[63]=
$$\left(2\,\text{Diff}\,\pi\left((A - \text{cfar}\,k)\,\xi + (-A + \text{cEqOut}\,k)\,R\left(\text{BesselI}\left[1, \frac{R}{\xi}\right]\text{BesselK}\left[0, \frac{L+R}{\xi}\right] + \text{BesselI}\left[0, \frac{L+R}{\xi}\right]\text{BesselK}\left[1, \frac{R}{\xi}\right]\right)\right)\right) \Big/ \left(k\,\xi\left(\text{BesselI}\left[0, \frac{L+R}{\xi}\right]\text{BesselK}\left[0, \frac{R}{\xi}\right] - \text{BesselI}\left[0, \frac{R}{\xi}\right]\text{BesselK}\left[0, \frac{L+R}{\xi}\right]\right)\right)$$

2D: Outside the droplet = Inside the shell when ceqout ~ cfar (when B → 0):

In[64]:=
```
ClearAll["Global`*"]
```

In[65]:=
```
$Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[65]= $\text{cEqOut} > 0 \;\&\&\; L > 0 \;\&\&\; \xi > 0 \;\&\&\; \text{Diff} > 0 \;\&\&\; R > 0$

Solve the stationary state equation for 2D droplets

In[66]:=
```
sol2DOutside[r_] =
 FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"] + A,
     c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[66]=
$$\frac{1}{4\,\text{Diff}\,\text{Log}\left[\frac{R}{L+R}\right]}\left((4\,\text{cEqOut}\,\text{Diff} - 4\,\text{cfar}\,\text{Diff} - A\,L\,(L + 2\,R))\,\text{Log}[r] + (4\,\text{cfar}\,\text{Diff} + A\,(L - r + R)\,(L + r + R))\,\text{Log}[R] + (-4\,\text{cEqOut}\,\text{Diff} + A\,(r - R)\,(r + R))\,\text{Log}[L + R]\right)$$

Check the boundary conditions

In[67]:=
```
FullSimplify@sol2DOutside[R]
```

Out[67]= cEqOut

In[68]:=
```
FullSimplify@sol2DOutside[R + L]
```

Out[68]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[69]:=
```
FullSimplify[Diff * Laplacian[sol2DOutside[r], {r, ϕ}, "Polar"] + A]
```

Out[69]= 0

Calculate surface area integrated fluxes at the droplet surface

In[70]:=
```
FluxOutside2D =
  FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutside[r], r] /. r → R]]
```

Out[70]=
$$A \pi R^2 + \frac{\pi\ (-4\ \text{cEqOut Diff} + 4\ \text{cfar Diff} + A\ L\ (L + 2\ R))}{2\ \text{Log}\left[\frac{R}{L+R}\right]}$$

2D: Outside the droplet = Inside the shell (Passive droplet):

In[71]:=
```
ClearAll["Global`*"]
```

In[72]:=
```
$Assumptions = cEqOut > 0 && L > 0 && Diff > 0 && R > 0
```

Out[72]= cEqOut > 0 && L > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 2D droplets

In[73]:=
```
sol2DOutsidePassive[r_] =
  FullSimplify[c[r] /. First@DSolve[{0 == Diff * Laplacian[c[r], {r, ϕ}, "Polar"],
       c[R] == cEqOut, c[R + L] == cfar}, c, r]]
```

Out[73]=
$$\frac{(\text{cEqOut} - \text{cfar})\ \text{Log}[r] + \text{cfar Log}[R] - \text{cEqOut Log}[L + R]}{\text{Log}\left[\frac{R}{L+R}\right]}$$

Check the boundary conditions

In[74]:=
```
FullSimplify@sol2DOutsidePassive[R]
```

Out[74]= cEqOut

In[75]:=
```
FullSimplify@sol2DOutsidePassive[R + L]
```

Out[75]= cfar

Check the solution by plugging it in the Diffusion equation

In[76]:=
```
FullSimplify[Diff * Laplacian[sol2DOutsidePassive[r], {r, ϕ}, "Polar"]]
```

Out[76]= 0

Calculate surface area integrated fluxes at the droplet surface

In[77]:=
```
FluxOutside2DPassive =
  FullSimplify[FullSimplify[-2 π * R * Diff * D[sol2DOutsidePassive[r], r] /. r → R]]
```

Out[77]=
$$\frac{2\ (-\text{cEqOut} + \text{cfar})\ \text{Diff}\ \pi}{\text{Log}\left[\frac{R}{L+R}\right]}$$

In[78]:=

1D: Inside the droplet:

```
In[79]:= ClearAll["Global`*"]

In[80]:= $Assumptions = Diff > 0 && k > 0 && cZeroIn > 0 && cEqIn > 0 && R > 0 && ξ > 0;
```

Solve the stationary state equation for 1D droplets

```
In[81]:= sol1DInside[x_] = FullSimplify[c[x] /. First@DSolve[
           {0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + ScZeroIn - k * (c[x] - cZeroIn),
            Derivative[1][c][0] == 0, c[R] == cEqIn}, c, x] /. Diff → k * (ξ^2)]
```

Out[81]= $$\frac{\text{cZeroIn}\,k + \text{ScZeroIn} - (-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{Cosh}\left[\frac{x}{\xi}\right]\,\text{Sech}\left[\frac{R}{\xi}\right]}{k}$$

Check the boundary conditions

```
In[82]:= FullSimplify@sol1DInside[R]

Out[82]= cEqIn

In[83]:= Limit[D[sol1DInside[x], x], x → 0]

Out[83]= 0
```

Check the solution by plugging it in the Reaction-Diffusion equation

```
In[84]:= FullSimplify[
          FullSimplify[Diff * Laplacian[sol1DInside[x], {x}, "Cartesian"] + ScZeroIn -
             k * (sol1DInside[x] - cZeroIn)] /. Diff → k * (ξ^2)]

Out[84]= 0
```

Calculate surface area integrated fluxes at the droplet surface

```
In[85]:= FluxInside1D = FullSimplify[
           FullSimplify[-2 * Diff * D[sol1DInside[x], x] /. x → R] /. Diff → k * (ξ^2)]
```

Out[85]= $$2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{Tanh}\left[\frac{R}{\xi}\right]$$

In[86]:=
```
FluxInside1DperVolume[R_] = FullSimplify[FluxInside1D / (2 * R)]
```

Out[86]= $$\frac{(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\xi\,\text{Tanh}\left[\frac{R}{\xi}\right]}{R}$$

In[87]:=
```
Normal[Series[FluxInside1DperVolume[R], {R, 0, 4}]]
```

Out[87]= $$\left(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn} + \frac{2\,R^4\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{15\,\xi^4} - \frac{R^2\,(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})}{3\,\xi^2}\right)$$

Check if ξ -> ∞ and k -> 0 goes back to FluxInside = ScZeroIn

In[88]:=
```
temp[F_] = FullSimplify[FullSimplify[FluxInside1D / (2 * R)] /. R → F * ξ]
```

Out[88]= $$\frac{(-\text{cEqIn}\,k + \text{cZeroIn}\,k + \text{ScZeroIn})\,\text{Tanh}[F]}{F}$$

In[89]:=
```
Limit[FullSimplify[Limit[temp[F], F → 0]], k → 0]
```

Limit: Warning: Assumptions that involve the limit variable are ignored.

Out[89]= ScZeroIn

1D: Outside the droplet = Inside the shell:

In[90]:=
```
ClearAll["Global`*"]
```

In[91]:=
```
$Assumptions = k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```

Out[91]= k > 0 && cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

We solve for A and B as s(c_{out}) at r = R = sOutceqout and s(c_{out}) at r = R + L = sOutcfar

In[92]:=
```
eq1 = A - k * cEqOut == sOutceqout;
eq2 = A - k * cfar == sOutcfar;
NSolve[{eq1, eq2}, {A, k}]
```

Out[94]= $$\left\{\left\{A \to \frac{1.\times(1.\,\text{cfar sOutceqout} - 1.\,\text{cEqOut sOutcfar})}{-1.\,\text{cEqOut} + 1.\,\text{cfar}},\; k \to -\frac{1.\times(1.\,\text{sOutceqout} - 1.\,\text{sOutcfar})}{1.\,\text{cEqOut} - 1.\,\text{cfar}}\right\}\right\}$$

In[95]:= $$A \to \frac{(\text{cfar sOutceqout} - \text{cEqOut sOutcfar})}{-\text{cEqOut} + \text{cfar}}$$

Out[95]= $$A \to \frac{\text{cfar sOutceqout} - \text{cEqOut sOutcfar}}{-\text{cEqOut} + \text{cfar}}$$

In[96]:=
```
k → -(sOutceqout - sOutcfar)/(cEqOut - cfar)
```
Out[96]= $k \to -\frac{\text{sOutceqout} - \text{sOutcfar}}{\text{cEqOut} - \text{cfar}}$

Solve the stationary state equation for 1D droplets

In[97]:=
```
sol1DOutside[x_] = FullSimplify[
  c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + A - k * c[x],
      c[R] == cEqOut, c[R + L] == cfar}, c, x] /. Diff → k * ξ^2]
```
Out[97]= $$\frac{e^{L/\xi}\left(-1+\text{Coth}\left[\frac{L}{\xi}\right]\right)\left(A\,\text{Sinh}\left[\frac{L}{\xi}\right]+(A-\text{cfar}\,k)\,\text{Sinh}\left[\frac{R-x}{\xi}\right]+(-A+\text{cEqOut}\,k)\,\text{Sinh}\left[\frac{L+R-x}{\xi}\right]\right)}{k}$$

Check the boundary conditions

In[98]:=
```
FullSimplify@sol1DOutside[R]
```
Out[98]= cEqOut

In[99]:=
```
FullSimplify@sol1DOutside[R + L]
```
Out[99]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[100]:=
```
FullSimplify[FullSimplify[Diff * Laplacian[sol1DOutside[x], {x}, "Cartesian"] +
    A - k * sol1DOutside[x]] /. Diff → k * ξ^2]
```
Out[100]= 0

Calculate surface area integrated fluxes at the droplet surface

In[101]:=
```
FluxOutside1D = FullSimplify[-2 * Diff * D[sol1DOutside[x], x] /. x → R]
```
Out[101]= $$-\frac{2\,\text{Diff}\left(-A+\text{cfar}\,k+(A-\text{cEqOut}\,k)\,\text{Cosh}\left[\frac{L}{\xi}\right]\right)\text{Csch}\left[\frac{L}{\xi}\right]}{k\,\xi}$$

1D: Outside the droplet = Inside the shell when ceqout ~ cfar (when B → 0):

In[102]:=
```
ClearAll["Global`*"]
```

In[103]:=
```
$Assumptions = cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0
```
Out[103]= cEqOut > 0 && L > 0 && ξ > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 1D droplets

In[104]:=
```
sol1DOutside[x_] = FullSimplify[
  c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"] + A,
      c[R] == cEqOut, c[R + L] == cfar}, c, x]]
```

Out[104]=
$$-\frac{((2\,\text{cfar}\,\text{Diff} + A\,L\,(L + R - x))\,(R - x)) + 2\,\text{cEqOut}\,\text{Diff}\,(L + R - x)}{2\,\text{Diff}\,L}$$

Check the boundary conditions

In[105]:=
```
FullSimplify@sol1DOutside[R]
```

Out[105]= cEqOut

In[106]:=
```
FullSimplify@sol1DOutside[R + L]
```

Out[106]= cfar

Check the solution by plugging it in the Reaction-Diffusion equation

In[107]:=
```
FullSimplify[Diff * Laplacian[sol1DOutside[x], {x}, "Cartesian"] + A]
```

Out[107]= 0

Calculate surface area integrated fluxes at the droplet surface

In[108]:=
```
FluxOutside1D =
 FullSimplify[FullSimplify[-2 * Diff * D[sol1DOutside[x], x] /. x → R]]
```

Out[108]=
$$\frac{2\,(\text{cEqOut} - \text{cfar})\,\text{Diff}}{L} - A\,L$$

1D: Outside the droplet = Inside the shell (Passive droplet):

In[109]:=
```
ClearAll["Global`*"]
```

In[110]:=
```
$Assumptions = cEqOut > 0 && L > 0 && l > 0 && Diff > 0 && R > 0
```

Out[110]= cEqOut > 0 && L > 0 && l > 0 && Diff > 0 && R > 0

Solve the stationary state equation for 1D droplets

In[111]:=
```
sol1DOutsidePassive[x_] =
 FullSimplify[c[x] /. First@DSolve[{0 == Diff * Laplacian[c[x], {x}, "Cartesian"],
      c[R] == cEqOut, c[R + L] == cfar}, c, x]]
```

Out[111]=
$$\frac{\text{cEqOut}\,(L + R - x) + \text{cfar}\,(-R + x)}{L}$$

Check the boundary conditions

In[112]:=
```
FullSimplify@sol1DOutsidePassive[R]
```
Out[112]=
```
cEqOut
```

In[113]:=
```
FullSimplify@sol1DOutsidePassive[R + L]
```
Out[113]=
```
cfar
```

Check the solution by plugging it in the Diffusion equation

In[114]:=
```
FullSimplify[Diff * Laplacian[sol1DOutsidePassive[x], {x}, "Cartesian"]]
```
Out[114]=
```
0
```

Calculate surface area integrated fluxes at the droplet surface

In[115]:=
```
FluxOutside1DPassive =
 FullSimplify[FullSimplify[-2 * Diff * D[sol1DOutsidePassive[x], x] /. x → R]]
```
Out[115]=

$$\frac{2\,(\text{cEqOut} - \text{cfar})\,\text{Diff}}{\text{L}}$$